



УДК 681.5

ПРИМЕР СОЗДАНИЯ ДИНАМИЧЕСКОГО ЗВЕНА НА ЯЗЫКЕ SCL Step 7

Левинский В.М., Степанов М.Т.
ОНАПТ

Аннотация: Показан учебный пример создания динамического звена на языке SCL в среде Step 7 с возможностью дальнейшего использования в контроллере SIMATIC S7-300

Annotation: Shows a case study of creating a dynamic link to the SCL language in Step 7 environment with the possibility of further use in the PLC SIMATIC S7-300

Ключевые слова: динамическое звено, язык SCL, среда Step 7, контроллер SIMATIC S7-300

Практически все среды программирования контроллеров включают в свой состав библиотеки логических и динамических звеньев (блоков), на базе которых можно быстро и эффективно создавать программы на языках международного стандарта IEC 1131-31. В литературе можно встретить достаточное количество обучающих примеров программ управления контроллерами. Значительно реже приводятся примеры создания самих звеньев, то есть программных кодов с комментариями, достаточными для понимания принципа работы звена.

Цель настоящей статьи – показать учебный пример создания простого динамического звена на языке SCL в среде Step 7 и возможность его дальнейшего использования в контроллере SIMATIC S7-300.

Рассмотрим инерционное звено первого порядка (апериодическое), входной $x(t)$ и выходной $y(t)$ сигналы которого связаны уравнением $T_{zv} dy(t)/dt + y(t) = k_{zv} x(t)$, где T_{zv} и k_{zv} – параметры звена.

Перейдем к дискретному представлению времени, воспользовавшись первой обратной разностью $dy(t)/dt \approx (y(nT_k) - y((n-1)T_k))/T_k$, где T_k – шаг дискретизации по времени. После подстановки этого приближенного равенства в исходное уравнение звена, можно получить рекуррентное уравнение, связывающее выходной сигнал $y(nT_k)$ со входным сигналом $x(nT_k)$ и предыдущим значением выходного сигнала $y((n-1)T_k)$:

$$y(nT_k) = k_1 x(nT_k) + k_2 y((n-1)T_k), \quad (1)$$

где $k_1 = k_{zv} T_k / (T_{zv} + T_k)$ и $k_2 = T_{zv} / (T_{zv} + T_k)$.

При $T_k < T_{zv}/10$ уравнение (1) с достаточной для практических применений точностью будет соответствовать исходному непрерывному звену. Осталось выполнить программную реализацию (1).

Для этого в среде программирования SIMATIC Manager создаем новый проект, например, под именем “aper_zveno”, и из меню “Insert” вставляем в него станцию “Station 300”. Далее в программе конфигурирования аппаратуры «HW Config» путем «перетаскивания» из каталога в состав станции добавляем контроллер S7-313C-2DP, модуль аналогового ввода/вывода SM 334 и коммуникационный процессор CP 343-1 Lean. Этих технических средств будет достаточно для проверки работоспособности цифровой реализации звена (1).

Из папки “Sources” проекта “aper_zveno” запускаем редактор языка SCL [1], в среде которого создаем программу, реализующую зависимость (1), как показано на рис. 1. Вначале следуют описания переменных, которые затем будут использоваться в программе. Следует обратить внимание, что переменная y_{n1} описывается, как статическая, значение которой не пропадает в промежутках между циклами работы контроллера.

Встроенные функции TIME_TO_DINT и DINT_TO_REAL используются для перевода значений шага квантования T_k и постоянной времени T_{zv} в тип REAL, что необходимо для дальнейших вычислений зависимости (1).

Промежуточные переменные k_1 и k_2 удобно вводить в рассмотрение, так как они упрощают представление окончательного результата. По известным значениям входного сигнала x_{inr} и выхода звена на предыдущем шаге вычислений y_{n1} рассчитывается значение y_{out} на текущем шаге, которое передается затем на выход звена. Последний оператор присваивает текущее значение выхода y_{out} предыдущему y_{n1} , которое запоминается до следующего шага вычислений.

После набора текста программы из меню “File” выполняем команду “Compile”, которая создает функциональный блок FB1. Данный блок следует вызывать в организационных блоках циклических прерываний по времени, например в OB35, так как они позволяют проводить расчеты с фиксированным шагом квантования по времени.



**АВТОМАТИЧНІ І АВТОМАТИЗОВАНІ СИСТЕМИ УПРАВЛІННЯ
ТЕХНОЛОГІЧНИМИ ПРОЦЕСАМИ**

```

FUNCTION_BLOCK FB1
VAR_input
  x_inp:REAL;//входной сигнал
  T_zv:TIME:= T#25S;//параметр звена
  k_zv:REAL:=1.0;//параметр звена
  Tk:TIME:= T#1S;//шаг квантования
END_VAR

VAR
  y_n1:REAL;//значение y на предыдущем шаге
END_VAR
VAR_TEMP
  k1,k2,rTk,rT_zv:REAL;//коэффициенты и переменные уравнения (1)
END_VAR
VAR_OUTPUT
  y_out:REAL;//выход звена
END_VAR
  rTk:=DINT_TO_REAL(TIME_TO_DINT(Tk));//перевод типа TIME в REAL
  rT_zv:=DINT_TO_REAL(TIME_TO_DINT(T_zv));
  k1:=rTk*k_zv/(rT_zv+rTk);//вычисление k1, k2
  k2:=rT_zv/(rT_zv+rTk);
  y_out:=k1*x_inp+k2*y_n1;//вычисление выхода звена согласно (1)
  y_n1:=y_out;//текущее значение выхода звена присваивается предыдущему
END_FUNCTION_BLOCK

```

Рис. 1 Текст программы, реализующей рекуррентные вычисления по зависимости (1) на языке SCL

На рис. 2 приведен пример использования функционального блока FB1 в организационном блоке OB35. Библиотечная функция FC105 “SCALE” позволяет перевести сигнал с физического аналогового входа PIW256, представленного в формате INT, в формат REAL промежуточной переменной MD20, которая затем поступает на вход FB1. Промежуточная переменная MD30 передаёт сигнал с выхода звена в формате REAL. Назначение библиотечной функции FC106 “UNSCALE” состоит в его дальнейшем преобразовании в тип INT, который может быть воспринят физическим аналоговым выходом PQW256 модуля аналогового ввода/вывода SM 334.




Рис. 2 Пример вызова функционального блока FB1 в организационном блоке OB35

Использование FB1 возможно только с экземплярным блоком данных DB1, который создаётся автоматически в ответ на приглашение ввести имя DB1.



**АВТОМАТИЧНІ І АВТОМАТИЗОВАНІ СИСТЕМИ УПРАВЛІННЯ
ТЕХНОЛОГІЧНИМИ ПРОЦЕСАМИ**

	Address	Declaration	Name	Type	Initial value	Actual value	Comment
1	0.0	in	x_inp	REAL	0.000000e+000	0.000000e+000	
2	4.0	in	T_zv	TIME	T#25S	T#25S	
3	8.0	in	k_zv	REAL	1.000000e+000	1.000000e+000	
4	12.0	in	Tk	TIME	T#1S	T#1S	
5	16.0	out	y_out	REAL	0.000000e+000	0.000000e+000	
6	20.0	stat	y_n1	REAL	0.000000e+000	0.000000e+000	

Рис. 3 Вид блока данных DB1

В блоке данных DB1 хранятся входные, выходные и статические переменные функционального блока FB1 в тактах между вычислениями.

Для проверки работоспособности вновь созданного динамического звена следует загрузить проект “aper_zveno” в контроллер S7-313C-2DP и наблюдать за изменениями выходного аналогового сигнала модуля SM334 при скачкообразном изменении входного. Как известно, в момент времени $t = T_{zv}$ значение на выходе $y(t) = 0,63 k_{zv} x(t)$.

Выводы: данный учебный пример показывает, как использовать теоретические знания по цифровым системам управления на практике при программировании реальных контроллеров. Аналогично могут создаваться и более сложные динамические звенья, в том числе и ПИД – регуляторы, рекуррентные зависимости для которых можно найти в [2].

Литература:

1. SIMATIC S7-SCL V5.1 для S7-300/S7-400 Руководство. Редакция 09/2000 A5E00059543-0. Siemens AG 2000. – 336 с.
2. Изерман Р. Цифровые системы управления: Пер. с англ. - М.: Мир, 1984. - 541 с.

УДК 681.513

ПОБУДОВА АЛГОРИТМУ РЕГУЛЮВАННЯ ХОЛОДОПРОДУКТИВНІСТЮ МУЛЬТИКОМПРЕСОРНОЇ ХОЛОДИЛЬНОЇ МАШИНИ

Заволінковська В. В., Гончаренко О. О.
Одеська національна академія харчових технологій

Анотація: В даній статті розглядається принцип побудови алгоритму регулювання холодопродуктивністю мультикомпресорної холодильної установки на теорії нечітких множин і нечіткої логіки.

Annotation: In this article the principle of algorithm development regulation multykompresornoyi cooling capacity of the refrigeration unit on the theory of fuzzy sets and fuzzy logic.

Ключові слова: нечіткий регулятор, алгоритм Сугено, АСР мультикомпресорної холодильної установки.

Дослідження, про які піде річ в статті, відносяться до області енергозберігаючих технологій і обладнання. На сьогоднішній день економія енергоресурсів і підвищення ефективності виробництва є одним із найпріоритетніших напрямків розвитку. Вирішення цієї проблеми в значній мірі допомагає підвищення якості регулювання технологічних процесів в реальних умовах експлуатації.

Регулювання холодопродуктивністю мультикомпресорної холодильної установки характеризуються значними зв'язками між окремими елементами, великою кількістю зовнішніх і внутрішніх впливів, частина з яких недоступна